

JEUX ELECTRONIQUES

BATTLEGAMES

Pour ZX Spectrum, ZX81, BBC, TRS-80, APPLE, VIC & PET



Hachette Jeunesse




DES PROGRAMMES
POUR VOTRE
MICRO



JEUX ÉLECTRONIQUES BATTEGAMES

par
D. Isaaman et J. Tyler

SOMMAIRE

- 
- 
- 
- 102130 2 A propos de ce livre
102560 4 Opération robots
102810 6 Le message codé
102810 8 Duel
1021240 10 Les blindés attaquent !
1021510 12 Le siège du
château fort
1021860 14 Les envahisseurs
1022220 16 L'arme secrète
1022600 18 Évadez-vous !
1022930 20 Le combat aérien *NE MARCHE PAS*
1023310 22 A bord d'un bombardier
supersonique
1024800 24 Bataille navale
sur l'Arctique
APRES GTO 26 Le mur du magicien
- 30 Missiles
antiaériens
version TRS-80
31 Missiles : version BBC
32 Missiles : version
ZX Spectrum
33 Missiles : version VIC 20
34 Missiles : version Apple
35 Missiles : version ZX81
36 Quelques trucs
pour aller plus loin
38 La programmation
est un jeu d'enfant
40 Petit lexique BASIC
46 Tableau de conversion
47 Réponses aux casse-tête

Conception graphique : G. Round et R. Priddy. Illustrations de R. Archer, J. Bamber, T. Baskeyfield, M. Newton et G. Round. L'édition française a été établie avec l'assistance technique de la Société European Media Business, 9, place des Ternes, 75017 Paris.

ÉCHOS/ÉLECTRONIQUE

HACHETTE

Avant-propos

Ce guide vous propose des programmes de jeux simples pour micro-ordinateurs. Écrits en BASIC (abréviation de l'anglais : *Beginner's All purpose Symbolic Instruction Code*), langage utilisé par la très grande majorité de ces matériels, ils sont prévus pour le ZX81, dont la mémoire n'est que de 1 K*. Ils s'adaptent facilement sur le ZX Spectrum, le BBC, le VIC 20, le TRS-80, l'Apple et le Pet grâce aux modifications signalées par des symboles renvoyant en bas de page.

Conçus pour tous, ces programmes n'exploitent peut-être que partiellement les possibilités de votre micro-ordinateur. Vous pourrez les modifier en conséquence, en vous reportant à la rubrique « Casse-Tête ». Vous trouverez également à la fin de ce manuel des idées, des trucs pour réaliser vos propres programmes, une table de conversion pour vous aider à les adapter sur votre machine ainsi qu'un lexique des termes BASIC les plus courants.

Comment entrer un programme

Lorsque des modifications dans les lignes de programme s'avéreront nécessaires pour des ordinateurs autres que le ZX81, vous rencontrerez les symboles suivants :

- ▲ VIC et Pet
- ★ BBC et Acorn Electron
- TRS-80
- Apple
- ZX Spectrum.

Chaque fois que vous rencontrerez le signe correspondant à votre ordinateur, consultez l'instruction indiquée et entrez-la à la place de celle prévue pour le ZX81.

**La capacité de mémoire correspondant à 1 024 caractères.
En moyenne, elle est de l'ordre de 16 à 64K.*

© Usborne Publishing Ltd, Londres, 1982.
© Hachette, Paris, 1983 pour l'édition française.



Mode d'emploi

- 1 Entrez les instructions telles qu'elles sont écrites ; soyez attentif à respecter ponctuation et espacements.
- 2 Appuyez, selon les modèles, sur la touche RETURN, NEWLINE, ou ENTER, à la fin de chaque ligne de programme.
- 3 Vérifiez soigneusement chaque ligne.
- 4 Assurez-vous qu'il n'y a ni confusion, ni omission de lignes de programmes ; n'hésitez pas à vous munir d'une feuille de papier et d'une règle.
- 5 Repérez les symboles et assurez-vous que vous enregistrez bien l'instruction correspondant à votre matériel.
- 6 Dans le cas du ZX81 ou du ZX Spectrum, les instructions ne doivent pas être entrées lettre à lettre ; utilisez la touche appropriée.

Si vous trouvez plus commode de vous faire dicter le programme, recommandez bien au lecteur de n'oublier ni virgules, ni points, ni parenthèses, ni espacements... Faites attention à bien différencier la lettre O et le chiffre 0.

Mise au point finale

Une fois entré, faites défiler votre programme sur l'écran en frappant l'instruction LIST, suivie éventuellement des numéros des lignes que vous désirez consulter (par ex. : LIST 20-100).




Vérifiez que tout a été convenablement tapé ; néanmoins il restera certainement quelques erreurs. Le mode d'emploi, fourni



par le fabricant de votre ordinateur, pourra vous aider à en corriger quelques-unes ; autrement, réécrivez la ligne en entier, l'ordinateur la mettra automatiquement à la place de l'ancienne.

Voici une liste des principales sources d'erreur :

- 1 Ligne oubliée.
- 2 Ligne mal numérotée.
- 3 Vous avez oublié d'aller à la ligne.
- 4 Virgules, points, parenthèses, deux points, points-virgules, espacements... omis ou confondus, tout particulièrement




dans les cas de programmes longs et complexes.

- 5 Vous avez utilisé une instruction inconnue de votre ordinateur.
- 6 Lettre O confondue avec 0 (zéro).
- 7 Erreur de chiffre ou nombre.

Mise en route

Pour lancer un jeu frappez l'instruction RUN. Attention, sur certains micros l'instruction RUN ne doit pas être utilisée dans les jeux faisant appel au hasard (Random number) ; faites GOTO à la place (GOTO le premier numéro du listing).

S'il subsiste une erreur de programmation, le jeu s'arrêtera en cours



d'exécution et l'ordinateur vous signalera, quelquefois, l'origine de cette interruption. Faites défiler à nouveau le programme et comparez-le scrupuleusement au listing de référence.

Si BREAK IN LINE 200 ou STOPPED AT 200 apparaît à la fin du jeu, relancez l'instruction de départ pour rejouer.

Un peu d'entraînement

Vous trouverez dans cet ouvrage des suggestions pour modifier ou compléter les jeux ; tentez également de les changer par vous-même. N'ayez pas peur, en aucun cas vous ne risquez d'endommager l'ordinateur ; et le programme d'origine reste disponible en cas d'échec.

Après quelques utilisations, vous souhaitez modifier la vitesse d'exécution du jeu ; vous trouverez un commentaire concernant la localisation et les modalités de cette intervention en bas de page*.

Là où figure l'instruction PRINT, vous avez la possibilité de modifier le texte entre guillemets. Sauf si votre appareil est un ZX81, vous pouvez même rajouter des messages. Choisissez un nombre (par exemple 105 pour glisser un texte entre 100 et 110), puis frappez PRINT et votre texte entre guillemets. Si votre ordinateur fait appel à des couleurs et des sons, aidez-vous de son mode d'emploi pour rendre encore plus intéressants les jeux présentés dans ce livre.



* Voir page 37 la note particulière concernant BBC et ZX Spectrum.



Opération robots

Nous sommes en 2582 et les peuples de la Terre sont en guerre contre les robots. Un de leurs missiles vient de tomber à proximité du quartier général de la Terre et menace de le faire sauter. Tous comptent sur vous pour déminer cet engin diabolique.

Vous devez trouver le code secret qui permettra cette opération, lettre code que connaît votre ordinateur. Entrez une lettre de votre choix et il vous indiquera si elle se trouve placée dans l'alphabet, avant ou après le bon code. Vous disposez de 4 essais avant que le missile n'explose.





Fonctionnement du programme

```

10 PRINT "OPERATION ROBOTS"
20 PRINT
30 PRINT "TROUVER LE BON CODE"
40 PRINT "LETTRE (A à Z) POUR"
50 PRINT "DESAMORCER LE MISSILE."
60 PRINT "VOUS AVEZ 4 CHANCES"
70 PRINT

```

Affichage du titre et des instructions de jeux.




```

★▲●80 LET C$ = CHR$ (37 + INT (RND *26 + 1))

```

Donne à la variable C\$ une valeur de lettre après conversion d'un chiffre aléatoire entre 0 et 26.



```

90 FOR G = 1 TO 4

```

Début d'une boucle : les lignes 100 à 140 se répètent 4 fois.

```

100 INPUT G$

```

Entrez votre choix en G\$

```

110 IF G$ = C$ THEN GOTO 210

```

Vérifie si votre lettre est celle du « code secret ». Si c'est le cas, l'ordinateur saute à 210.

```

120 IF G$ < C$ THEN PRINT "PLUS";
130 IF G$ > C$ THEN PRINT "MOINS";
140 PRINT "LOIN QUE "; G$;

```

Contrôle si votre lettre se trouve placée avant ou après celle du « code secret » et affiche le message approprié.

```

150 NEXT G

```



Fin de la boucle - Retourne ligne 90.

```

160 PRINT
170 PRINT "BOOOOOOOOMMM..."
180 PRINT "IL A EXPLOSE."
190 PRINT "LE BON CODE ETAIT "; C$
200 STOP
210 PRINT "TICK... FZZZZ... CLICK..."
220 PRINT "VOUS AVEZ REUSSI"
230 PRINT STOP

```

Affichage en cas d'échec à l'issue des 4 essais.

Affichage en cas de réussite.

Pour les autres matériels que le ZX81, réalisez les modifications d'instructions suivantes :

```

★▲●80 LET C$ = CHR$ (64 + INT (RND (1) *26 + 1))
■80 LET C$ = CHR$ (64 + INT (RND (0) *26 + 1))
s80 LET C$ = CHR$ (64 + INT (RND *26 + 1))

```

Complément au programme

Un message supplémentaire peut être affiché au dernier essai. Modifiez la ligne 220 en ajoutant un point-virgule et créez la ligne 225, comme ci-après :

```

220 PRINT "VOUS AVEZ RÉUSSI";

```

et ajoutez la nouvelle ligne 225

```

225 IF G = 4 THEN PRINT "DE JUSTESSE"

```

Casse-tête

Essayez de trouver un « truc » pour que le programme vous donne plus ou moins de chances de découvrir la lettre code.





Le message codé

Opérateur au centre des communications par laser, votre travail consiste à intercepter des messages émis par les robots et à les envoyer vers le quartier général. Un message codé, vital, est attendu, et de sa bonne transmission dépend la déroute des robots lors de leur prochaine attaque. Ce jeu va tester votre

capacité de mémorisation d'un groupe de lettres aperçues très brièvement. Dès que vous aurez indiqué la difficulté (4 à 10) choisie, un groupe de lettres apparaîtra pour quelques instants en haut et à gauche de votre écran. Tentez de vous en souvenir et frappez ces lettres sur le clavier. Ce message était-il le bon ?



Fonctionnement du programme

3	▲●10 CLS	Vide l'écran avant de démarrer le jeu.
	20 PRINT "LE MESSAGE CODE"	
	30 PRINT	
	40 PRINT "DIFFICULTE CHOISIE (4 à 10)"	Donne à D une valeur (de 4 à 10) correspondant à la difficulté du jeu.
	50 INPUT D	
	60 IF D<4 OR D>10 THEN GOTO 40	Vérifie que le nombre est bien compris entre 4 et 10.
	70 LET MS = " "	Crée une chaîne vide (« null string »), dénommée MS, dans laquelle l'ordinateur pourra stocker le message secret.
	80 FOR I = 1 TO D	
★▲●	90 LET MS = MS + CHR\$(INT(RND*26 + 38))	Le programme tourne D fois sur lui-même et choisit à chaque passage un caractère qu'il ajoute à la chaîne déjà créée et entrée en MS.
	100 NEXT I	
	▲●110 CLS	
	120 PRINT "ENVOYER CE MESSAGE : "	Vide l'écran et affiche le message.
	130 PRINT	
	140 PRINT MS	
★▲●	150 FOR I = 1 TO D*8	Le message reste inscrit sur l'écran le temps qu'il faut pour que les touches s'exécutent D × 8 fois.
	160 NEXT I	
	▲●170 CLS	Vide l'écran à la fin des boucles.
	180 INPUT N\$	Place votre version du message en N\$.
	190 IF N\$ = MS THEN GOTO 240	Vérifie que votre message est identique à MS et saute à 240 si c'est le cas.
	200 PRINT "MESSAGE ERRONE"	
	210 PRINT "VOUS AURIEZ DU ENVOYER : "	Affichage en cas d'erreur : communication du message d'origine que vous n'avez pu identifier.
	220 PRINT MS	
	230 GOTO 260	
	240 PRINT "MESSAGE CORRECT"	
	250 PRINT "LA GUERRE EST TERMINEE"	
	260 STOP	

Pour les autres matériels que le ZX81, réalisez les modifications d'instructions suivantes :

- 10, 110, 170 HOME
- ▲10, 110, 170 PRINT CHR\$(147)
- ★▲●90 LET MS=MS+CHR\$(INT(RND(1)*26+65))
- 90 LET MS=MS+CHR\$(INT(RND(0)*26+65))
- 90 LET MS=MS+CHR\$(INT(RND*26+65))
- ▲●150 FOR I=1 TO D*180
- ★150 FOR I=1 TO D*400



Le ZX81 a recours à un code différent de celui des autres ordinateurs (qui utilisent le code « ASCII ») pour l'accès aux caractères.

Pour rendre le jeu plus difficile

On peut modifier ce programme, en insérant dans le message secret des chiffres et des ponctuations :

- ZX90 LET MS = MS + CHR\$(INT(RND*43 + 21))
- ★▲●90 LET MS = MS + CHR\$(INT(RND(1)*43 + 48))
- 90 LET MS = MS + CHR\$(INT(RND(0)*43 + 48))
- 90 LET MS = MS + CHR\$(INT(RND*43 + 48))

Casse-tête



Trouvez le moyen d'afficher le message plus longtemps.

Duel

Placés dos à dos, vous et l'autre cow-boy, vous vous éloignez l'un de l'autre de 10 pas, puis lentement vous vous retournez et approchez votre main de la crosse du revolver. Serez-vous le plus rapide ? Tirerez-vous le premier ?

Le micro-ordinateur inscrira les nombres de 1 à 10 sur l'écran pour simuler les 10 pas, laissera passer un temps relativement bref et affichera : IL DÉGAINE... Vous devez alors appuyer sur n'importe quelle touche le plus vite possible. Si vous êtes assez rapide, votre vie sera sauve. N'appuyez pas sur une touche avant la visualisation du message : IL DÉGAINE, vous seriez déclaré perdant.

Pour modifier la vitesse du jeu

Le temps de réaction à l'affichage du message IL DÉGAINE peut être modifié en agissant sur le dernier nombre de la ligne 130. Un nombre plus petit diminue ce temps de réaction. (Pour BBC, modifiez le nombre entre parenthèses ligne 140.)

Pour rendre le jeu plus difficile

En changeant le programme, vous pouvez vous donner la possibilité de rater votre coup : ligne 140, 190 devient 220. Ajouter les lignes ci-contre :

61



```

10 CLS
20 PRINT "DUEL"
30 PRINT "VOUS ETES DOS A DOS"
40 PRINT "ELOIGNEZ-VOUS DE 10 PAS..."
50 FOR I=1 TO 10
60 PRINT I; "...";
70 NEXT I
80 PRINT
90 FOR I=1 TO RND*200
100 NEXT I

110 IF INKEYS<>" " THEN GOTO 160

120 PRINT "IL DEGAINE..."

130 FOR I=1 TO 5
140 IF INKEYS<>" " THEN GOTO 190
150 NEXT I
    
```



```

160 PRINT "ET TIRE."
170 PRINT "VOUS ETES MORT."
180 GOTO 210
190 PRINT "MAIS VOUS AVEZ TIRE LE PREMIER."
200 PRINT "IL EST MORT."
210 STOP
    
```

Pour les autres matériels que le ZX81, réalisez les modifications d'instructions suivantes :

```

10 HOME
10 PRINT CHR$(147)
60 PRINT I; "...";
65 FOR J=1 TO 300 : NEXT J
90 FOR I=1 TO RND(1)*1000
90 FOR I=1 TO RND(0)*1000
110 IF PEEK(-16384)>127 THEN GOTO 160
110 GET IS : IF IS<>" " THEN GOTO 160
110 IF INKEYS(1)<>" " THEN GOTO 160
130 FOR I=1 TO 20
130 FOR I=1 TO 50
DELETE 130, 150
140 IF PEEK(-16384)>127 THEN GOTO 190
140 GET IS : IF IS<>" " THEN GOTO 190
140 IF INKEYS(40)<>" " THEN GOTO 190
    
```

```

ZX81 220 IF RND>.3 THEN GOTO 190
220 IF RND(1)>.3 THEN GOTO 190
220 IF RND(0)>.3 THEN GOTO 190
230 PRINT "MAIS À COTÉ"
240 GOTO 90
    
```

Fonctionnement du programme

Boucle reproduite 10 fois et
qui provoque à chaque fois
l'affichage d'un nombre et
de deux points.

Boucle d'attente : l'ordinateur
« tourne sur lui-même » un nombre
de fois correspondant au
chiffre aléatoire
pris entre 0 et 200.

Vérifie que vous n'êtes pas
en train de tricher en appuyant
sur une touche avant l'affichage
du message **IL DÉGAINE.**

Affiche le signal du tir
(presser une touche).

Contrôle le clavier pour voir
si vous appuyez sur une touche.
Si c'est la cas il saute à 190.
Ce contrôle se trouve à
l'intérieur d'une boucle pour
répéter plusieurs fois cette
opération (ce qui vous donne
plus de chance d'intervenir
à temps).


Affichage correspondant à
un échec
(si vous avez été
touché ou si vous avez triché).

Affiche votre réussite.

Les boucles
d'attente peuvent être
écrites sur une seule ligne,
sauf dans le cas du ZX81
(voir ligne 65).



Boucle d'attente supplémentaire
pour les ordinateurs plus rapides.



Pour vérifier si une ou
plusieurs touches du cla-
vier ont été pressées,
différentes instructions
existent : GET pour le VIC 20,
PEEK pour Apple et INKEY
pour les autres appareils.



Casse-tête

Pouvez-vous également
donner à l'ordinateur
la possibilité de rater
son coup ?

Solution page 47



Les blindés attaquent !

Sous vos ordres, une brigade de blindés a été envoyée en plein désert pour détruire la vieille forteresse en ruine où se cache la dernière base importante des forces robots hors des E.R.U.* Il ne reste plus que 5 missiles à bord de votre tank : ne gaspillez plus vos munitions, ajustez au mieux la direction et la hausse de votre canon à chaque tir. L'ordinateur vous demande de lui communiquer l'angle de tir compris entre -90° (extrême gauche) et $+90^{\circ}$ (extrême droite) ainsi que la hausse du tir, de 0 (au sol) à 90° (au zénith), qui détermine la distance du point d'impact. Avez-vous les nerfs assez solides pour accomplir cette mission ?

*E.R.U. : États Robots Unis.



10



10 PRINT "LES BLINDES ATTAQUENT"
 ★■▲●20 LET T=INT(RND*181)-90

★■▲●30 LET D=RND

40 FOR G=1 TO 5
 50 PRINT "DIRECTION (-90 à 90) ?"
 60 INPUT T1
 70 PRINT "HAUSSE (0 à 90) ?"
 80 INPUT B
 90 LET D1=SIN(2*(B/180*3.1416))

100 IF ABS(T-T1)<2 AND ABS(D-D1)<.05 THEN GOTO 220



La fonction ABS permet de ne considérer que la valeur absolue d'un nombre ; elle ignore donc les signes + et -.

110 PRINT "MISSILE ENVOYE TROP"
 120 IF T1<T THEN PRINT "A GAUCHE"
 130 IF T1>T THEN PRINT "A DROITE"

140 IF ABS(D1-D)>.05 AND T1<>T THEN PRINT "ET"



Un point-virgule à la fin d'une instruction PRINT indique à l'ordinateur qu'il ne devra pas aller à la ligne lorsqu'il rencontrera la prochaine instruction PRINT.

150 IF D-D1>.05 THEN PRINT "TROP PRES";
 160 IF D1-D>.05 THEN PRINT "TROP LOIN";
 170 PRINT
 180 NEXT G
 190 PRINT "MALHEUR - VOUS AVEZ ECHOUÉ"
 200 PRINT "ALLEZ CACHER VOTRE HONTE"
 210 STOP
 220 PRINT "BOOOUUUMMM..."
 230 PRINT "VOUS AVEZ FAIT MOUCHE"
 240 STOP

Pour les autres matériels que le ZX81, réalisez les modifications d'instructions suivantes :

★▲●20 LET T=INT(RND(1)*181)-90
 ■20 LET T=INT(RND(0)*181)-90
 ★▲●30 LET D=RND(1)
 ■30 LET D=RND(0)



L'instruction LET n'est nécessaire que pour le ZX81 et le ZX Spectrum.

Affichage du message correspondant (suite du précédent) : TROP PRES ou TROP LOIN.

Affiche votre échec.

Affiche votre réussite.

Fonctionnement du programme

Sélection d'un nombre entier compris entre - 90 et + 90 (direction du tir).

Sélection d'un nombre entre 0 et 1 pour définir à quelle distance se trouve le château.

Donne aux variables A et B les valeurs de votre choix en direction et hausse du canon.

Calcule, à partir de la hausse choisie, la distance d'impact du missile (réponse entre 0 et 1).

Si votre direction (à 2 degrés près) et votre distance (à 0,05 m près) sont conformes aux valeurs définies aléatoirement lignes 20 et 30, vous avez touché le château. Le programme saute ligne 220 pour affichage.

Si non conforme, compare votre direction et celle définie ligne 20 et affiche le message correspondant.

Si la distance d'impact ne correspond pas (comparaison avec ligne 30) affichage du message "ET".

Casse-tête

Pouvez-vous donner aux robots la possibilité de vous voir et de vous détruire avant épuisement de vos munitions ?

Solution page 47

Le siège du château fort

Une bataille furieuse et sanglante oppose le roi et son plus mortel ennemi, le Baron félon. Vous êtes un des meilleurs archers royaux et, caché derrière des buissons qui entourent le château ennemi, vous décochez vos flèches dès l'apparition des assiégés aux créneaux.

Votre ordinateur affiche une rangée de huit points et la lettre O. Vous devez très rapidement presser la bonne touche, de 1 à 9 selon la position de O, avant que la lettre ne disparaisse.

Combien d'hommes du Baron allez-vous réussir à éliminer ?

131

10 PRINT "LE SIEGE DU CHATEAU FORT"
20 LET S=0
30 FOR G=1 TO 10
40 LET RS=""
★▲●50 LET T=INT (RND*9+1)
60 FOR L=1 TO 9
70 IF L=T THEN LET RS=RS+"O"
80 IF L<>T THEN LET RS=RS+"."
90 NEXT L
100 PRINT RS,
★▲●110 FOR I=1 TO 12
★▲●120 LET IS=INKEY\$
130 IF VAL ("O"+IS)=T THEN GOTO 170
140 NEXT I
150 PRINT "RATÉ"
160 GOTO 190
170 PRINT "MOUCHE"
180 LET S=S+1
190 NEXT G
200 PRINT "VOUS AVEZ REUSSI"; S; "COU"
210 STOP

Pour les autres matériels que le ZX81,
réalisez les modifications d'instructions suivantes :

■50 LET T=INT(RND(0)*9+1)
/★▲●50 LET T=INT(RND(1)*9+1)
▲110 FOR I=1 TO 200
■110 FOR I=1 TO 60
/★▲●110 FOR I=1 TO 90
●115 LET IS=""
★120 LET IS=INKEY\$(1)
●120 IF PEEK(-16384)>127 THEN GET IS
▲120 GET IS

Pour modifier la vitesse du jeu

Si vous trouvez que le jeu se déroule soit trop lentement, soit trop rapidement, il vous sera facile d'en modifier la vitesse en intervenant sur le dernier chiffre de la ligne 110. Plus le nombre est petit, plus le jeu est rapide.

Casse-tête

Êtes-vous capable de changer le programme pour avoir 2 cibles :

S, un des soldats du Baron, comptant pour 1 point. C, un des chevaliers félons, comptant pour 5 points.

Solution page 47

Fonctionnement du programme

Mise à zéro du score pour démarrer.

Début d'une boucle qui vous donne 10 essais.

Crée une chaîne libre RS.

Donne à T une valeur comprise entre 1 et 9.

L'ordinateur fait 9 tours, ajoutant à chaque passage un caractère à la chaîne RS. La position de l'ennemi (la lettre O) est donnée par la valeur de T ; dans les autres cas nous trouverons des points.

Affiche la chaîne de caractères.

Boucle (12 fois) pendant laquelle l'ordinateur vérifie si une touche est pressée. (Fin de la boucle ligne 140).

Vérifie si vous appuyez sur la bonne touche : si oui, saute ligne 170.

Affiche "RATÉ" si vous avez appuyé sur une mauvaise touche ou bien trop tard. Saute ensuite à 190.

Augmente votre score de 1 point.

Retour ligne 30 pour un nouvel essai.

SUR 10 — Affiche votre score à l'issue des 10 essais.

Regardez comment fonctionne le score dans ce programme (lignes 20 et 180). Vous pouvez adapter ce principe à d'autres programmes.



Les envahisseurs

Une multitude de robots envahisseurs s'apprête à vous attaquer. Heureusement, votre panoplie d'armes est très étendue, car vous aurez besoin d'un outil de guerre spécifique pour détruire chaque catégorie de robots.

Chaque robot sera représenté à l'écran par un symbole. Appuyez rapidement sur la touche correspondante et vérifiez combien d'envahisseurs vous avez réussi à éliminer. Attention : dans certains cas, vous devez également utiliser la touche SHIFT.



161

10 PRINT "LES ENVAHISSEURS"

20 LET H=0

30 FOR T=1 TO 25

★ ■ ▲ ● 40 FOR I=1 TO INT(RND*30+20)
50 NEXT I

★ ■ ▲ ● 60 LET A=INT(RND*20)

★ ■ ▲ ● 70 LET D=INT(RND*15)

★ ■ ▲ ● 80 LET P\$=CHR\$(INT(RND*53+11))

▲ ● 90 CLS

100 FOR J=0 TO D

110 PRINT

120 NEXT J

130 PRINT TAB(A); P\$

★ ■ ▲ ● 140 FOR I=1 TO 15

★ ■ ▲ ● 150 LET R\$=INKEY\$

160 IF R\$=P\$ THEN GOTO 210

170 IF R\$<>" THEN GOTO 190

180 NEXT I

190 PRINT "MANQUE"

200 GOTO 230

210 PRINT "UN DE MOINS"

220 LET H=H+1

230 NEXT T

▲ ● 240 CLS

250 PRINT "VOUS TOTALISEZ "; H; " POINTS"

260 STOP

Fonctionnement
du programme

Remise à zéro du score.

Début d'une boucle
donnant 25 essais.

Boucle créant un délai aléatoire.

Sélection de nombres pour dé-
finir une position horizontale
et verticale sur l'écran.

Choisit un caractère.

Vide l'écran.


Déplace le curseur vers le bas
de l'écran, ligne après ligne,
jusqu'à la ligne n° D
(déjà choisie ligne 70).

Déplace le curseur de A
espacements vers la droite de
l'écran et affiche ici le caractère
choisi ligne 80.

Vérifie plusieurs fois (15)
que vous appuyez bien sur
une touche, s'assure qu'il
s'agit de la bonne touche
et saute ensuite jusqu'à
la ligne correspondante
pour donner le résultat.

Augmente le score de 1 point.
Retourne au point de départ de
la boucle pour un nouvel essai
(sauf au 25°).

Affiche le score obtenu
en fin de partie.



Pour les autres matériels que le ZX81, réalisez les modifications d'instructions suivantes :

- ★▲●40 FOR I=1 TO INT(RND(1)*300+200)
- 40 FOR I=1 TO INT(RND(0)*300+200)
- ★▲●60, 70 changer RND en RND(1)
- 60, 70 changer RND en RND(0)
- ★▲●80 LET PS=CHR\$(INT(RND(1)*58+33))
- 80 LET PS=CHR\$(INT(RND(0)*58+33))
- 80 LET PS=CHR\$(INT(RND*58+33))
- 90, 240 HOME
- ▲90, 240 PRINT CHR\$(147)
- ★■▲●140 FOR I=1 TO 150
- 145 RS="--"
- ★150 RS=INKEY\$(1)
- ▲150 GET RS
- 150 IF PEEK(-16384) > 127 THEN GET RS

Pour modifier la vitesse du jeu

Au fur et à mesure que vous maîtriserez ce jeu, accélérez-en la cadence. Pour cela diminuez la valeur du dernier nombre de la ligne 140.

Le curseur

Votre micro a peut-être recours à une instruction plus simple pour positionner le curseur à l'écran (lignes 100 à 130). Vérifiez sur son mode d'emploi.

Casse-tête

Voici les robots les plus dangereux.



U

Horrible



V

Sans-merci



W

Scorpion



X

Tête-en-fer



Y

Regard-d'acier

Vous est-il possible de modifier le programme pour que la destruction d'un "super-robot" vous rapporte 100 points au lieu de 10 ?

Solution page 47

L'arme secrète

Si vous étiez en mesure de détruire le Magasin des Pièces de Rechange des robots, enterré dans les étendues désertiques de l'Est des E.R.U.*, l'attaque des robots – imminente –, serait sérieusement paralysée.

Votre nouvelle arme secrète, encore inconnue des robots, peut traverser la roche sans bruit, désagrégeant tout sur son passage. Mais le Magasin est si intelligemment dissimulé que vous ne pouvez guère diriger votre arme qu'à l'aveuglette en espérant arriver le plus près possible de l'objectif.

Définissez la difficulté du jeu (4 étant le plus petit nombre autorisé) et communiquez les coordonnées (X et Y) probables de l'objectif. Vous entrerez séparément X et Y (appuyez sur RETURN, NEWLINE ou ENTER à chaque fois). Pour connaître les valeurs possibles de X et Y, reportez-vous au listing du programme.

* États Robots Unis.



20

1K

Fonctionnement du programme

```

▲●10 CLS
20 PRINT "L'ARME SECRETE"
30 PRINT "CHOIX DE DIFFICULTE"
40 INPUT D
50 IF D<4 THEN GOTO 30

```

Donne à D la valeur que vous avez choisie (niveau de difficulté) et vérifie qu'elle est supérieure ou égale à 4.

```

*■▲●60 LET X=INT(RND*D+1)
*■▲●70 LET Y=INT(RND*D+1)
80 FOR G=1 TO D+5

```

Sélection des valeurs de X et Y.

Début d'une série de boucles qui vous donne un nombre équivalent d'essais (selon la difficulté choisie).

```

90 PRINT "COORDONNEES X ET Y"
100 INPUT X1
110 INPUT Y1

```

Donne à X1 et Y1 les valeurs de votre choix.

```

120 LET Z=SQR((X-X1)*(X-X1)+(Y-Y1)*(Y-Y1))

```

Place en Z le résultat du calcul de la distance entre votre tir et l'objectif.



SQR permet l'extraction de la racine carrée.

```

130 IF Z=0 THEN GOTO 200
140 IF Z<3 THEN PRINT "PAS LOIN"
150 IF Z>3 THEN PRINT "MAUVAISE LOCALISATION"
160 NEXT G

```

Vérifie la valeur de Z. Êtes-vous suffisamment proche de l'objectif ? Affichage de la réponse (si Z=0, vous avez gagné !).

Fin de la boucle. Retour pour essai suivant.

```

170 PRINT "LES ROBOTS VOUS ONT REPERE"
180 PRINT "VOUS ETES M... AGGHHH"
190 STOP

```

Affichage en cas d'échec complet (à la fin des essais).

```

200 PRINT "VOUS L'AVEZ DETRUIT EN "
210 PRINT G;" COUPS"
220 STOP

```

Affichage en cas de réussite « VOUS L'AVEZ DETRUIT EN N COUPS ».

Pour les autres matériels que le ZX81, réalisez les modifications d'instructions suivantes :

```

●10 HOME
▲10 PRINT CHR$(147)
*▲●60 LET X=INT(RND(1)*D+1)
■60 LET X=INT(RND(0)*D+1)
*▲●70 LET Y=INT(RND(1)*D+1)
■70 LET Y=INT(RND(0)*D+1)

```

Casse-tête

Peut-on modifier la comptabilisation du score pour que vous obteniez 1 point à chaque fois que vous échouez de peu et 10 points en cas de réussite complète ?



Solution page 47

Évadez-vous !

Les robots vous ont capturé, désarmé et enfermé. Votre cas semble désespéré, mais, tout à coup, vous vous rappelez l'existence de la montre-sonar toujours à votre poignet. Une fréquence bien précise peut neutraliser vos gardes en les faisant entrer en résonance ; mais attention, une fréquence trop basse peut faire vibrer les murs qui s'écrouleraient sur vous ; trop haute, elle se traduirait par un mal de tête insupportable.

Saurez-vous échapper à l'horreur des prisons des robots ? Pour connaître la gamme des fréquences utilisables, consultez attentivement le programme.

23



```

▲●10 CLS
20 PRINT "EVADEZ-VOUS !"
★▲●30 LET F=INT(RND*100+1)

40 LET L=1
50 LET H=1

60 FOR G=1 TO 5

70 PRINT "VOTRE FREQUENCE ? ";
80 INPUT F1

90 IF ABS(F-F1)<5 THEN GOTO 290

100 IF F-F1>40 THEN GOTO 170
110 IF F1-F>40 THEN GOTO 230

120 PRINT "SANS EFFET"
130 NEXT G
140 PRINT "TROP TARD."
150 PRINT "LA FREQUENCE ETAIT "; F
160 STOP

170 IF L=2 THEN GOTO 210
180 PRINT "TROP BASSE... ATTENTION !"
190 LET L=2
200 GOTO 130
210 PRINT "LES MURS S' ECROULENT..."
220 STOP

230 IF H=2 THEN GOTO 270
240 PRINT "TROP ELEVEE... AIEEEE"
250 LET H=2
260 GOTO 130
270 PRINT "VOUS VOUS EVANOUISSEZ"
280 STOP
290 PRINT "EVASION REUSSIE"
300 STOP
    
```

Fonctionnement du programme

Donne à F la valeur d'un nombre entre 1 et 100 (fréquence des robots).

Donne à L et H la valeur 1 si vous tirez trop bas ou trop haut (voir ligne 170 à 190 et 230 à 250).

Début d'une boucle qui vous donne 5 essais.

Demande votre choix et donne à F1 cette valeur.

Contrôle si votre choix se trouve compris entre + 5 ou - 5 par rapport à F.

Si c'est le cas, saute à 290 pour afficher « ÉVASION RÉUSSIE ».

Passe à 170 si votre choix est inférieur de 40 à F.

Passe à 230 si votre choix est supérieur de 40 à F.

Vous indique (affichage) que votre choix est à moins de 40 et plus de 5 de F et retourne au point de départ de la boucle. Au dernier tour, affichage des résultats et de la réponse.


Contrôle la valeur de L. Au premier passage L=1 (ligne 40) ; l'ordinateur affiche l'avertissement (ligne 180), puis modifie la valeur L (L=2 ligne 190) et renvoie au tour suivant. En cas de deuxième passage (L=2), l'ordinateur va directement à 210 et affiche l'échec. Arrêt du jeu.

Même chose, avec H entre les lignes 230 et 280.

Pour les autres matériels que le ZX81, réalisez les modifications suivantes :

```

●10 HOME
▲10 PRINT CHR$(147)
★▲●30 LET F=INT(RND(1)*100+1)
■30 LET F=INT(RND(0)*100+1)
    
```



Casse-tête

Imaginons que vos trois gardes possèdent chacun leur propre fréquence de résonance. Comment modifier le programme pour trouver les trois fréquences nécessaires au succès de votre évasion ?

Solution page 48



Pour rendre le jeu plus difficile

Deux possibilités :
réduire la marge d'erreur autour de la fréquence à découvrir
– réduction de 5, ligne 90 –, ou bien augmenter l'amplitude des fréquences au-delà de 100, ligne 30.

Le combat aérien

Vous êtes face au pirate du ciel ; vous allez devoir tenter de le placer dans votre ligne de tir et de l'abattre avant qu'il ne le fasse lui-même. S'il se porte devant vous, accélérez ; s'il se place derrière vous, ralentissez.

Votre ordinateur vous communiquera vos vitesse et positions par rapport au pirate. Dès le début du jeu, soyez prêt à appuyer sur les touches appropriées : A pour accélérer, R pour ralentir jusqu'à la bonne position pour faire feu (touche F).



26

```

▲●10 CLS
20 PRINT "COMBAT AERIEN"
★▲●30 LET V=INT (RND*11-5)

★▲●40 LET S=-INT (RND*3+1)

50 IF ABS (S)>20 THEN GOTO 230
▲●60 CLS

70 PRINT "VOUS ETES ";
80 IF S<0 THEN PRINT "DERRIERE LUI"
90 IF S>0 THEN PRINT "DEVANT LUI"
100 IF S=0 THEN PRINT "EN POSITION DE TIR"
110 PRINT "VOTRE VITESSE EST ";
120 IF V>0 THEN PRINT "TROP ELEVEE"
130 IF V<0 THEN PRINT "TROP FAIBLE"
140 IF V=0 THEN PRINT "CONVENABLE"

★▲●150 LET I$=INKEY$
160 IF I$="A" THEN LET V=V+1
170 IF I$="R" THEN LET V=V-1
180 IF I$="F" AND S=0 THEN GOTO 250
    
```

Fonctionnement du programme

Détermine pour V une valeur comprise entre - 5 et + 5, soit votre vitesse relative par rapport au pirate.

Donne à S une valeur négative ; ceci définit la distance vous séparant du pirate. Au départ, vous vous situez derrière lui.

Vérifie que cette distance n'est pas supérieure à 20. Si c'était le cas, il saute à 230 pour vous indiquer « VOUS L'AVEZ PERDU DE VUE ».

Contrôle les valeurs de S et V et affiche votre position et votre vitesse vis-à-vis du pirate.

Êtes-vous en train d'appuyer sur une touche et laquelle ? Touche A, votre vitesse augmente de 1 ; touche R, votre vitesse diminue de 1. Si vous pressez la touche F et que S = 0 (ligne 100), l'ordinateur saute à 250. Si S est différent de 0, alors rien ne se passe.

190 LET S=S+V	Calcule la nouvelle distance qui sépare les deux avions.
★ ■ ▲ ● 200 FOR I=1 TO 20	Boucle d'attente.
210 NEXT I	
220 GOTO 50	Retour ligne 50 pour recommencer.
230 PRINT "VOUS L' AVEZ PERDU DE VUE"	
240 GOTO 330	
250 IF ABS (V) < 2 THEN GOTO 290	Si vous ouvrez le feu dans de bonnes conditions (ligne 180), vérifie votre vitesse : si elle est inférieure à 2 (en valeur absolue) il saute à 290 ; dans le cas contraire, il passe à 260.
260 PRINT "VOUS L' AVEZ"	
270 PRINT "JUSTE EFFLEURE"	Si la vitesse est supérieure à 2 au moment du tir, affiche le message « VOUS L' AVEZ JUSTE EFFLEURE » et retourne ligne 40 pour définir une nouvelle valeur de S au prochain essai.
280 GOTO 40	
★ ■ ▲ ● 290 IF RND > . 7 THEN GOTO 320	Effectue une opération aléatoire pour voir si le pirate a tiré avant vous.
300 PRINT "VOUS L' AVEZ ABATTU !"	
310 GOTO 330	
320 PRINT "VOUS ETES TOUCHE"	Affiche le message correspondant au résultat de la ligne 290.
330 STOP	

Pour les autres matériels que le ZX81, réalisez les modifications d'instructions suivantes :

● 10, 60 HOME	● 145 I\$=""
▲ 10, 60 PRINT CHR\$ (147)	★ 150 I\$=INKEY\$ (1)
★ ▲ ● 30 LET V=INT (RND (1)*11-5)	● 150 IF PEEK (-16384) > 127 THEN GET I\$
■ 30 LET V=INT (RND (0)*11-5)	▲ 150 GET I\$
★ ▲ ● 40 LET S=-INT (RND (1)*3+1)	★ ■ ▲ ● 200 FOR I=1 TO 200
■ 40 LET S=-INT (RND (0)*3+1)	■ 290 IF RND (0) > . 7 THEN GOTO 320
	★ ▲ ● 290 IF RND (1) > . 7 THEN GOTO 320

Pour rendre le jeu plus facile

Si ce jeu vous semble trop difficile, ajoutez les trois lignes suivantes qui vous permettront de visualiser les deux tirs sur l'écran.

195 IF ABS(S)>10 THEN GOTO 200

196 PRINT TAB (W/2) ; "LUI"

197 PRINT TAB (S+W/2) ; "VOUS"

W = nombre de caractères par ligne (sur votre ordinateur).

Casse-tête

L'opération aléatoire, ligne 290, est programmée en votre faveur. Transformez-la pour que vos chances de gagner soient identiques à celles du pirate.



Solution page 48

A bord d'un bombardier supersonique

En mission au-dessus
du territoire des robots, à bord de
votre bombardier supersonique, votre ordinateur
de bord affiche 5 graphiques établis à partir de
photographies infrarouges relayées par satellite
qui montrent la répartition des bataillons de
robots. Vos réserves de carburant ne vous
permettent d'atteindre qu'un seul objectif ; faites
rapidement votre choix en vue de détruire le plus
grand nombre de robots et larguez une de vos
bombes « à corrosion » (elles contiennent une
substance très active capable de détruire la
carcasse d'un robot en quelques secondes).

Pour larguer la bombe, entrez le nombre affiché
à côté du graphique représentant la plus grande
densité de robots. Si deux graphiques sont
identiques, choisissez le plus petit nombre
affiché. Serez-vous décoré à votre retour de
mission ?



30



```

10 CLS
20 PRINT "BOMBARDIER SUPERSONIQUE"
30 DIM B(5)
40 LET S=0

```



Fonctionnement du programme

Cette instruction transforme la variable B en un « tableau variable » qui peut contenir simultanément cinq valeurs différentes, repérées comme B(1), B(2), B(3), B(4), B(5). Remet le score à 0.

```
50 FOR G=10 TO 1 STEP -1
```

Crée une boucle de dix essais en ordre décroissant (de 10 à 1 en retirant 1 à chaque fois).

Vous disposez ainsi d'un temps plus long au premier essai (boucle d'attente ligne 190 qui est fonction de G).

```
60 LET M=1
```



Donne la valeur 1 au numérotage du graphique de densité des robots.

```

70 FOR I=1 TO 5
80 LET B(I)=INT(RND*10+1)
90 IF B(I)>B(M) THEN LET M=I
100 NEXT I

```

Choisit cinq nombres et les place dans le tableau B(1) à B(5). Recherche le plus grand et établit la correspondance avec M.

```

110 CLS
120 FOR I=1 TO 5
130 PRINT I ;
140 FOR J=1 TO B(I)
150 PRINT "*" ;
160 NEXT J
170 PRINT
180 NEXT I

```

Affiche les cinq nombres sous forme de rangées d'étoiles (*).

```

190 FOR I=1 TO G*3
200 LET IS=INKEY$
210 IF IS<>" " THEN GOTO 270
220 NEXT I
230 PRINT "TROP TARD"

```

Vérifie que vous pressez une touche et saute à 270 si c'est le cas.

Si vous n'avez pas appuyé à temps sur le clavier, affiche le message « TROP TARD » et déclenche un nouvel essai.

```

240 FOR J=1 TO 10
250 NEXT J
260 GOTO 280
270 IF VAL(IS)-M THEN LET S=S+1
280 NEXT G

```

Contrôle si la touche pressée était la bonne et compte 1 point en cas de réussite.

Nouvel essai.

```

290 PRINT "VOUS AVEZ PLACE "; S ; " BOMBES"
300 PRINT "SUR LA BONNE CIBLE"
310 IF S=10 THEN PRINT "VOUS ETES UN HEROS"
320 IF S<10 THEN PRINT "RESULTATS MEDIOCRES"
330 STOP

```

Affiche votre score après le dixième essai.

Pour les autres matériels que le ZX81, réalisez les modifications d'instructions suivantes :

●10, 110 HOME	●195 IS=""
▲10, 110 PRINT CHR\$(147)	★200 IS=INKEY\$(1)
★▲●80 LET B(I)=INT(RND(1)*10+1)	▲200 GET IS
■80 LET B(I)=INT(RND(0)*10+1)	●200 IF PEEK(-16384)>127 THEN GET IS
★■▲●190 FOR I=1 TO G*30	★■▲●240 FOR J=1 TO 400

Pour modifier la vitesse du jeu

Pour vous donner plus de chance, augmentez la valeur du dernier nombre, ligne 190. Diminuez cette valeur au fur et à mesure que votre adresse croît ; jusqu'où irez-vous ?

Casse-tête

Proposez une solution pour augmenter le nombre des cibles potentielles.



Réponse page 48

Bataille navale sur l'Arctique

Sans armes et gravement endommagé, votre bateau tente de rentrer à bon port, louvoyant dans une mer encombrée d'icebergs. Un navire ennemi vous talonne. Mais, curieusement, s'il peut vous détecter, les icebergs, par contre, lui sont invisibles.

Votre dernière chance consiste à l'obliger à en percuter un. Votre ordinateur affiche une grille sur laquelle apparaissent votre position (Y), celle du navire ennemi (Z) et l'emplacement des icebergs (*). Vos déplacements se font espace par espace au nord, sud, est ou ouest, alors que votre poursuivant se déplace par la route la plus directe (y compris en diagonale). S'il parvient à vous rattraper (une des 8 positions environnantes), vous êtes capturé et si, dans votre fuite, vous heurtez un iceberg, votre bateau coule. Arriverez-vous à bon port ?



40

```

10 PRINT "BATAILLE NAVALE"
20 DIM B(8,8)

★ ■ ▲ ● 30 LET N=INT(RND*8+4)

★ ■ ▲ ● 40 FOR I=1 TO N
★ ■ ▲ ● 50 LET B(INT(RND*8+1),INT(RND*8+1))=23
★ ■ ▲ ● 60 NEXT I

★ ■ ▲ ● 70 LET SX=INT(RND*8+1)
★ ■ ▲ ● 80 LET SY=INT(RND*8+1)
★ ■ ▲ ● 90 IF B(SX,SY) < > 0 THEN GOTO 70
★ ■ ▲ ● 100 LET B(SX,SY)=63

★ ■ ▲ ● 110 LET YX=INT(RND*8+1)
★ ■ ▲ ● 120 LET YY=INT(RND*8+1)
★ ■ ▲ ● 130 IF B(YX,YY) < > 0 THEN GOTO 110
★ ■ ▲ ● 140 LET B(YX,YY)=62

★ ■ ▲ ● 150 CLS
★ ■ ▲ ● 160 FOR Y=1 TO 8
★ ■ ▲ ● 170 FOR X=1 TO 8
★ ■ ▲ ● 180 IF B(X,Y)=0 THEN GOTO 210
★ ■ ▲ ● 190 PRINT CHR$(B(X,Y));
★ ■ ▲ ● 200 GOTO 220
★ ■ ▲ ● 210 PRINT " ";
★ ■ ▲ ● 220 PRINT " ";
★ ■ ▲ ● 230 NEXT X
★ ■ ▲ ● 240 PRINT
★ ■ ▲ ● 250 NEXT Y
    
```

Fonctionnement du programme

Ceci crée la grille. B forme un tableau dont les dimensions sont 8 x 8.

N représente le nombre d'icebergs : il varie de 4 à 11 selon le résultat de la valeur aléatoire.

Le code pour le caractère * : 23 pour le ZX81 et 42 pour les autres ordinateurs. Le programme tourne N fois, plaçant les étoiles * à des positions aléatoires sur la grille.

Place le navire ennemi sur la grille pour le départ du jeu [code 63 (ZX81) ou 90 pour le caractère Z] en vérifiant que la position n'est pas déjà occupée par un iceberg.

Même chose pour votre navire (code 62 ou 89 pour le caractère Y).

Affecte la grille avec position des icebergs et des bateaux.

Casse-tête



Essayez d'augmenter la taille de la grille, puis le nombre des icebergs.

Solution page 48

260 LET B(YX, YY) = 0

Efface votre position actuelle pour vous permettre d'en définir une nouvelle.

270 PRINT "DIRECTION (N, S, E, O) :"
280 INPUT DS

Donne à DS la valeur d'une des 4 lettres choisies pour indiquer votre déplacement.

★ ■ ▲ {
290 LET YY = YY + (DS = "S" AND YY < > 8)
300 LET YY = YY - (DS = "N" AND YY < > 1)
310 LET YX = YX + (DS = "E" AND YX < > 8)
320 LET YX = YX - (DS = "O" AND YX < > 1)

Calcule votre nouvelle position et contrôle que vous n'êtes pas sorti des limites de la grille.

★ ■ ▲ ● 330 IF B(YX, YY) = 63 THEN GOTO 500
★ ■ ▲ ● 340 IF B(YX, YY) = 23 THEN GOTO 600

Vérifie que cette position ne vous fait pas heurter un iceberg ou entrer en contact avec le navire ennemi. Si c'est le cas, saute à 500 ou 600 pour donner le résultat.

★ ■ ▲ ● 350 LET B(YX, YY) = 62
360 LET B(SX, SY) = 0

Vous vous êtes déplacé sans problème : voici votre position.

370 LET SX = SX + SGN(YX - SX)
380 LET SY = SY + SGN(YY - SY)

Calcule la nouvelle position du navire ennemi.

★ ■ ▲ ● 390 IF B(SX, SY) = 62 THEN GOTO 500
★ ■ ▲ ● 400 IF B(SX, SY) = 23 THEN GOTO 700

Vérifie cette nouvelle position : vous a-t-il attrapé ou a-t-il heurté un iceberg ? Saute à 500 ou 700 si c'est le cas.

★ ■ ▲ ● 410 LET B(SX, SY) = 63

Fixe sa position.

420 GOTO 150
500 PRINT "VOUS ETES PRIS"
510 GOTO 800
600 PRINT "VOUS COULEZ"
610 GOTO 800
700 PRINT "SAUVE : IL A HEURTE UN ICEBERG"
800 STOP

Fin de boucle : retour à la ligne 150 pour un nouveau tour.

Pour les autres matériels que le ZX81, réalisez les modifications d'instructions suivantes :

★ ■ ▲ ● 30, 50, 70, 80, 110, 120 changer RND en RND(1)
■ 30, 50, 70, 80, 110, 120 changer RND en RND(0)
★ ■ ▲ ● 50, 340, 400 changer 23 en 42
★ ■ ▲ ● 100, 330, 410 changer 63 en 90
★ ■ ▲ ● 140, 350, 390 changer 62 en 89
▲ 150 PRINT CHR\$(147) ;
● 150 HOME
★ ■ ▲ {
290 LET YY = YY - (DS = "S" AND YY < > 8)
300 LET YY = YY + (DS = "N" AND YY < > 1)
310 LET YX = YX - (DS = "E" AND YX < > 8)
320 LET YX = YX + (DS = "O" AND YX < > 1)

Les ordinateurs peuvent décider si une expression est mauvaise ou bonne. Dans le premier cas, ils lui donnent la valeur 0 et dans le second, selon les appareils, la valeur +1 ou -1. C'est la raison pour laquelle les lignes 290 à 320 sont susceptibles d'être modifiées.



Le mur du magicien

Le mur que le magicien a construit pour encercler son bastion secret n'est pas un mur ordinaire. Ses pierres ne sont autres que les victimes pétrifiées par la célèbre colère du magicien ; de plus, il peut se déplacer. Si vous réussissez à créer une brèche suffisante dans ce mur à l'aide de vos fidèles catapultes qui jettent des boulets gigantesques, vous romprez l'enchantement et les victimes retrouveront leur forme originelle. Mais attention, le magicien prépare un nouveau charme qui peut arrêter les projectiles en pleine course et les renvoyer à leur expéditeur... et cela marche quelquefois !

En cas d'échec, vous ne serez qu'une pierre de plus dans ce mur diabolique.

```
▲●10 CLS
20 PRINT "LE MUR DU MAGICIEN"
30 PRINT
40 PRINT "SOUHAITEZ-VOUS UN PEU D'AIDE ?"
50 INPUT IS
★■▲●60 IF IS(1)="0" THEN GOSUB 740
70 PRINT "DIFFICULTE DE JEU"
80 PRINT "5=FACILE, 1=DIFFICILE"
90 INPUT Q
100 DIM W(8,8)
110 DIM E(8)
120 FOR Y=1 TO 8
130 FOR X=1 TO 4
★■▲●140 LET W(X,Y)=20
150 NEXT X
160 NEXT Y
170 LET Z=0
180 FOR C=1 TO 3
190 LET D=INT(RND*80+21)
200 GOSUB 430
210 IF Z=1 THEN GOTO 1150
220 PRINT "VOUS VOUS TROUVEZ A " ; D ; " METRES"
★■▲●230 LET W=INT(RND*41)-20
240 IF W=0 THEN PRINT "PAS DE VENT"
250 IF W<0 THEN PRINT "VENT VERS LA DROITE"
260 IF W>0 THEN PRINT "VENT VERS LA GAUCHE"
270 PRINT "DONNEZ LA HAUSSE (1 à 90)"
280 INPUT A
290 PRINT "DONNEZ LA VITESSE"
```

```

300 INPUT V
310 LET A=A/180*3.1416
320 LET H=TAN(A)*(D-W)-(5*(D-W)^2)/(V*COS(A))^2
330 LET H=INT(H/Q)
340 IF H>0 AND H<9 THEN GOSUB 540
350 IF H<1 THEN PRINT "TIR TROP COURT"
360 IF H>8 THEN PRINT "TIR TROP LONG"
*■▲●370 IF RND>.2 THEN GOTO 200
380 PRINT "LE MUR A BOUGE..."
390 NEXT C
400 PRINT "VOUS AVEZ ETE CHANGE"
410 PRINT "EN PIERRE"
420 STOP
*■▲●430 FOR I=1 TO 30
440 NEXT I
▲●450 CLS
460 PRINT
470 FOR Y=8 TO 1 STEP-1
480 FOR X=1 TO 8
*■▲●490 PRINT CHR$(W(X,Y));
500 NEXT X
510 PRINT
520 NEXT Y
530 RETURN
540 GOSUB 650
550 LET W(E(H)-1,H)=0
560 IF V*COS(A)>50 THEN LET W(E(H)-2,H)=0
*■▲●570 IF RND>.5 AND H<1 THEN LET W(E(1),1)=29
*■▲●580 IF RND>.5 AND H>5 AND H<8 THEN LET W(E(H+1)-1,H+1)=0
590 IF H>1 THEN GOTO 630
600 FOR Y=2 TO 8
*■▲●610 IF RND<.5 THEN LET W(E(Y)-1,Y)=0
620 NEXT Y
630 GOSUB 650
640 RETURN
650 FOR Y=1 TO 8
660 LET X=1
670 IF W(X,Y)=0 OR X=8 THEN GOTO 700
680 LET X=X+1
690 GOTO 670
700 LET E(Y)=X
710 IF X=1 THEN LET Z=1

```

Le mur du magicien (suite)

```
720 NEXT Y
730 RETURN
740 PRINT "VOUS ATTAQUEZ LE"
750 PRINT "DERNIER BASTION DU"
760 PRINT "FAMEUX MAGICIEN QUI"
770 PRINT "SE CACHE DERRIERE UN"
780 PRINT "INTERMINABLE MUR DE PIERRE"
790 PRINT "CONSTITUE DE SES"
800 PRINT "PRECEDENTES VICTIMES."
810 PRINT "VOUS ETES LE SEUL A"
820 PRINT "POUVOIR ATTAQUER CE"
825 PRINT "MUR ET ROMPRE CET"
830 PRINT "ENCHANTEMENT."
840 PRINT "POUR DETRUIRE CE MUR"
850 PRINT "VOUS DISPOSEZ DE"
860 PRINT "CATAPULTES, MAIS,"
870 PRINT "ATTENTION, LE MAGICIEN"
880 PRINT "A LE POUVOIR DE FAIRE"
890 PRINT "AVANCER OU RECULER LE"
895 PRINT "MUR, ET EVENTUELLEMENT,"
900 PRINT "DE VOUS RETOURNER"
910 PRINT "VOS PROPRES COUPS."
920 GOSUB 1110
930 PRINT "APRES CHAQUE TIR, VOUS"
940 PRINT "VOYEZ UNE COUPE"
950 PRINT "DU MUR QUI VOUS MONTRE"
960 PRINT "L' ETENDUE DES"
970 PRINT "DOMMAGES."
980 PRINT "NOTEZ QUE CERTAINS"
990 PRINT "BOULETS PROVOQUENT"
1000 PRINT "PLUS DE DEGATS, ET"
1010 PRINT "QUE, PLUS ILS SE"
1020 PRINT "DEPLACENT VITE"
1030 PRINT "HORIZONTALEMENT, PLUS"
1040 PRINT "ILS SONT DANGEREUX."
1050 PRINT "SAUREZ-VOUS VAINCRE"
1060 PRINT "LE MAGICIEN A TEMPS"
```

```

1070 PRINT "POUR SAUVER LES MILLIERS"
1080 PRINT "D'AMES PRISONNIERES ?"
1090 GOSUB 1110
1100 RETURN
1110 PRINT "APPUYEZ SUR UNE TOUCHE " ;
★▲●1120 IF INKEYS="" THEN GOTO 1120
▲●1130 CLS
1140 RETURN
1150 PRINT "VOUS AVEZ REUSSI A"
1160 PRINT "FAIRE UN TROU DANS LE MUR"
1170 PRINT "DU MAGICIEN, VOUS AVEZ"
1180 PRINT "VAINCU SES POUVOIRS ET"
1190 PRINT "DELIVRE SES VICTIMES."
1200 STOP

```

Pour les autres matériels que le ZX81,
réalisez les modifications
d'instructions suivantes :

```

■changer tous les RND en RND(0)
★▲●changer tous les RND en RND(1)
●10,450,1130 HOME
▲10,450,1130,PRINT CHR$(147) ;
★■▲●60 IF LEFT$(IS,1)="0" THEN GOSUB 740
★■▲●140 LET W(X,Y)=29
■▲●430 FOR I=1 TO 200
★430 FOR I=1 TO 1000
★■▲●490 PRINT CHR$(W(X,Y)+32) ;
★1120 IS=GET$
●1120 GET IS
▲1120 GET IS : IF IS="" THEN GOTO 1120

```

Modifications à apporter au jeu

Vous pouvez rendre le mur plus épais en modifiant
les lignes suivantes :


Ligne 100 : augmentez la valeur du premier chiffre (ex : 10).
Ligne 130 : retirez 4 au nombre choisi ligne 100 et placez-le
à la fin de la ligne (ex : 6).

Ligne 480 : modifiez le dernier nombre comme ligne 100
par le même chiffre (ex : 10).

Si vous désirez plus de temps, augmentez la valeur
du dernier nombre ligne 180.

Missiles antiaériens

Ce jeu est différent des autres, car il utilise le mode graphique. Chaque ordinateur ayant des instructions graphiques très personnelles, nous vous proposons plusieurs listings de programmes. Lisez les informations ci-contre et recherchez le programme correspondant à votre ordinateur.



Missiles antiaériens : version TRS-80

```
10 CLS
20 DIM Y(3), F(3)
30 N=1
40 PS=INT(RND(0)*3+1)
50 P=INT(RND(0)*36+5)
60 GOSUB 400
70 FOR I=PS TO 100 STEP PS
80 GOSUB 300
90 F$=INKEY$
100 IF F$="" OR N>3 THEN 130
110 F(N)=1
120 N=N+1
130 FOR J=1 TO 3
140 RESET (32 * J, 46-Y(J))
150 IF F(J)=0 OR Y(J)>45 THEN 190
160 Y(J)=Y(J)+1
170 IF POINT (32 * J, 46-Y(J)) THEN 230
180 SET (32 * J, 46-Y(J))
190 NEXT
200 NEXT
210 PRINT @0, "RATE"
220 END
230 PRINT @0, "TOUCHE"
240 END
300 RESET (I-PS, P) : RESET (I-PS+1, P)
310 SET (I, P) : SET (I+1, P)
```

Comment jouer

Vous disposez de trois bases de missiles, chacune capable d'en lancer un. Quand un avion approche, estimez son altitude et sa vitesse et ouvrez le feu coup par coup.

Les missiles sont lancés par simple pression sur une touche quelconque du clavier, dans l'ordre suivant : gauche, milieu et droite.

Combien d'avions ennemis abattrez-vous ?

Missiles antiaériens : version BBC

```
10 MODE 5
20 VDU 23, 224, 224, 160, 144, 144, 143, 128, 128, 255
30 VDU 23, 225, 0, 0, 0, 0, 240, 12, 2, 255
40 VDU 23, 226, 16, 56, 84, 16, 16, 16, 0, 0
50 VDU 23, 227, 0, 0, 0, 8, 8, 8, 8, 60
60 DIM Y(3), F(3)
70 N=1 : MS=16
80 PS=RND(20)+10
90 P=RND(500)+400
100 PROCDISPLAY
110 FOR I=PS TO 1100 STEP PS
120 PROCPLANE(I-PS, P, 0) : PROCPLANE(I, P, 3)
130 FS=INKEY$(0)
140 IF FS="" OR N>3 THEN 170
150 F(N)=TRUE
160 N=N+1
170 FOR J=1 TO 3
180 IF NOT F(J) THEN 240
190 PROCMISSILE(J, 0)
200 Y(J)=Y(J)+MS
210 IF Y(J)<1024 THEN PROCMISSILE(J, 3)
220 X=J * 320-I : Y=Y(J)+32-P
230 IF X<128 AND X>-40 AND Y>-32 AND Y<2 THEN 280
240 NEXT
250 NEXT
260 CLS : PRINT "RATE"
270 END
280 PROCPLANE(I, P, 1) : SOUND 0, -15, 5, 20
290 END
300 DEF PROCPLANE(X, Y, C)
310 GCOL 0, C
320 MOVE X, Y
330 VDU 5, 224, 225, 4
340 ENDPROC
350 DEF PROCMISSILE(N, C)
360 GCOL 0, C
370 MOVE 320 * N, 32+Y(N)
380 VDU 5, 226, 4
390 ENDPROC
400 DEF PROCDISPLAY
410 FOR I=1 TO 3
420 MOVE I* 320, 32
430 VDU 5, 227, 8, 226, 4
440 NEXT
450 ENDPROC
```

Missiles antiaériens : version ZX Spectrum

```

10 CLS
15 INVERSE 0
20 DIM y(3)
30 DIM f(3)
40 LET n=1
50 LET ms=8
60 LET ps=INT(RND * 6 +5)
70 LET p=INT(RND * 140+20)
80 GOSUB 400
90 FOR i=ps TO 240 STEP ps
100 LET c=1 : LET x=i-ps
110 GOSUB 300
120 LET c=0 : LET x=i
130 GOSUB 300
140 LET F$=INKEY$
150 IF F$=" " OR N>3 THEN GOTO 170
160 LET f(n)=1 : LET n=n+1
170 FOR j=1 TO 3
180 LET c=1 : GOSUB 350
190 IF f(j)=0 OR y(j)>148 THEN GOTO 240
200 LET y(j)=y(j)+ms
210 LET c=0 : GOSUB 350
220 LET x=j* 64-i : LET y=p-Y(j)
230 IF x>-1 AND x<12 AND y<10 AND y>-5 THEN GOTO 280
240 NEXT j
250 NEXT i
260 PRINT AT 0,0 ; "RATE"
270 STOP
280 PRINT AT 0,0 ; "TOUCHE"
290 STOP
300 INVERSE c
310 PLOT x,p
320 DRAW 0,8 : DRAW 3,-6
330 DRAW 8,0 : DRAW 2,-2
340 DRAW -13,0 : RETURN
350 INVERSE c
360 PLOT 64 *j+4,Y(j)+4
370 DRAW 0,6 : DRAW -2,-2
380 DRAW 2,2 : DRAW 2,-2
390 RETURN
400 FOR a=65 TO 66
410 FOR b=0 TO 7
420 READ c
430 POKE USR CHR$(a)+b,c
440 NEXT b
450 NEXT a
460 FOR j=1 TO 3
470 PRINT AT 21,8*j ; "[AB]"
480 NEXT j
490 RETURN
500 DATA 0,0,0,0,255,255,
127,63
510 DATA 0,252,252,252,
255,255,254,252

```

Missiles antiaériens : version VIC 20

```
10 PRINT CHR$(147)CHR$(5) ;
20 POKE 36879,8
60 DIM Y(3),F(3)
70 N=1 : MS=2
90 P=INT(RND(1)*9+2)*2
110 FOR I=1 TO 21 STEP RND(1)/2+.5
120 GOSUB 300
130 GET F$
140 IF F$=" " OR N>3 THEN 170
150 F(N)=-1
160 N=N+1

170 FOR J=1 TO 3
180 IF F(J)=0 THEN 220
190 POKE 8164+J*5-Y(J)*22,32
200 IF Y(J)=22 THEN 240
210 Y(J)=Y(J)+1
220 POKE 8164+J*5-Y(J)*22,30
230 IF ABS(I-J*5)<=1 AND P=Y(J) THEN 280
240 NEXT
250 NEXT
260 PRINT CHR$(147) ; "RATE"
270 END
280 PRINT CHR$(147) ; "TOUCHE"
282 POKE 36877,220 : POKE 36878,15
284 FOR K=1 TO 500 : NEXT
286 POKE 36877,0 : POKE 36878,0
290 END
300 POKE 8163+I-P*22,32
310 POKE 8164+I-P*22,121
320 RETURN
```

Note aux utilisateurs de ZX81 et ZX Spectrum.


Pour faire apparaître des caractères graphiques dans une instruction PRINT, se reporter à la convention suivante : le caractère [signifie appuyez sur la touche GRAPHICS. Puis frappez les touches suivantes (avec SHIFT pour ZX81) et pressez à nouveau GRAPHICS quand vous rencontrez le caractère].

Missiles antiaériens : version APPLE

```
10 HOME
20 HGR
30 HCOLOR=3
40 DIM Y(3), F(3)
50 N=1 : MS=5
60 PS=INT(RND(1)*6+4)
70 P=INT(RND(1)*135+11)
80 GOSUB 400
90 FOR I=PS TO 265 STEP PS
100 X=I-PS : Y=159-P : C=0 : GOSUB 300
110 X=I : C=3 : GOSUB 300
120 F$="" : IF PEEK(-16384)>127 THEN GET F$
130 IF F$="" OR N>3 THEN 160
140 F(N)=1
150 N=N+1
160 FOR J=1 TO 3
170 C=0 : GOSUB 350
180 IF F(J)=0 OR Y(J)>145 THEN 230
190 Y(J)=Y(J)+MS
200 C=3 : GOSUB 350
210 X=J*70-I : Y=P-Y(J)
220 IF X>-1 AND X<15 AND Y>-9 AND Y<5 THEN 270
230 NEXT
240 NEXT
250 VTAB 22 : PRINT "RATE"
260 END
270 VTAB 22 : PRINT "TOUCHE"
280 END
300 HCOLOR=C
310 HPLLOT X, Y TO X, Y-8
320 HPLLOT TO X+3, Y-2 : HPLLOT TO X+12, Y-2
330 HPLLOT TO X+14, Y : HPLLOT TO X, Y
340 RETURN
350 HCOLOR=C
360 HPLLOT 70 *J, 158-Y(J) TO 70 *J, 154-Y(J)
370 RETURN
400 FOR J=1 TO 3
410 HPLLOT 70 *J-5, 159 TO 70 *J+5, 159
420 NEXT
430 RETURN
```

Missiles antiaériens : version ZX81

```
10 CLS
20 DIM Y(3)
30 DIM F(3)
40 LET N=1
50 LET P=INT(RND*19+2)*2
60 FOR I=1 TO 48
70 PLOT I,P
80 PLOT I+1,P
90 UNPLOT I-1,P
100 LET B$=INKEYS
110 IF B$="" OR N>3 THEN GOTO 140
120 LET F(N)=1
130 LET N=N+1
140 FOR J=1 TO 3
150 IF F(J)=0 THEN GOTO 190
160 UNPLOT J*16,Y(J)
170 IF Y(J)=42 THEN GOTO 210
180 LET Y(J)=Y(J)+1
190 PLOT J*16,Y(J)
200 IF P=Y(J) AND (J*16=I OR J*16=I+1)
THEN GOTO 260
210 NEXT J
220 NEXT I
230 CLS
240 PRINT "RATE"
250 STOP
260 CLS
270 PRINT "TOUCHE"
280 STOP
```




Quelques trucs pour aller plus loin

Voici quelques idées pour compléter les programmes de ce livre ou vos propres programmes. Dans la plupart des cas vous ne pourrez pas les utiliser pour le ZX81, puisque les jeux occupent à eux seuls la totalité de sa mémoire (1K) ; mais ce n'est guère le cas pour les autres micro-ordinateurs.

Attention : vous serez peut-être limité par les espacements laissés libres entre les numéros de lignes de programmes. N'hésitez pas à renuméroter la liste sans oublier de modifier également les numéros de lignes auxquels renvoient les instructions GOTO et GOSUB.

Pour chacun des programmes, l'ordinateur peut afficher une série d'informations vous rappelant le mode d'emploi du jeu.



Il suffit de rajouter quelques lignes au début de la liste et de placer le sous-programme à la fin.

```
10 PRINT "TITRE DU JEU"  
11 PRINT "VOULEZ-VOUS CONNAITRE"  
12 PRINT "LE MODE D'EMPLOI ?"  
15 INPUT IS  
S ZX17 IF IS (1) = "0" THEN GOSUB 1000  
★ ■ ▲ ● 17 IF LEFTS (IS, 1) = "0" THEN GOSUB 1000
```

Emplacement du programme principal

```
1000 PRINT "MODE D'EMPLOI"  
1010 PRINT .....  
1999 RETURN
```

A priori, vous n'êtes pas limité par le nombre des écritures. Numérotez bien chaque ligne, frappez PRINT et vérifiez que le nombre de caractères entre guillemets ne dépasse pas les possibilités d'affichage de votre ordinateur.

L'instruction RETURN doit obligatoirement figurer à la fin du sous-programme. Dans le cas contraire, il ne fonctionnerait pas.

L'ordinateur peut s'arrêter et vous attendre



Si vos lignes d'instructions sont particulièrement nombreuses et que vous ne voulez pas les voir disparaître prématurément de l'écran, le sous-programme que nous vous proposons maintenant présente un réel intérêt : il arrête le déroulement du programme en un point défini tant qu'une touche du clavier n'a pas été frappée. Insérez l'instruction GOSUB à l'endroit choisi et écrivez ce sous-programme :

```
1000 PRINT "POUR CONTINUER"  
1005 PRINT "FRAPPEZ UNE TOUCHE"  
S ■ ZX1010 IF INKEYS="" THEN GOTO 1010  
★ 1010 GET IS  
● 1010 GET IS : IF IS="" THEN GOTO 1010  
▲ 1020 PRINT  
1030 RETURN
```

L'ordinateur vous parle



L'ordinateur peut vous poser des questions et agir, ensuite, en fonction de votre réponse. Dans le premier exemple ci-dessous, il refusera de jouer avec ceux dont l'initiale du prénom ne sera pas J.

```
1 PRINT "QUEL EST TON PRENOM ?"  
2 INPUT IS  
S ZX3 IF IS(1) <> "J" THEN GOTO 1000  
★ ■ ▲ ● 3 IF LEFTS (IS, 1) <> "J" THEN GOTO 1000  
4 PRINT "VOUS POUVEZ JOUER"  
5 PRINT "ETES-VOUS PRET ?"  
6 INPUT JS  
S ZX7 IF JS(1) <> "0" THEN GOTO 5  
★ ■ ▲ ● 7 IF LEFTS (JS, 1) <> "0" THEN GOTO 5
```

Programme principal

```
1000 PRINT "DESOLÉ, CE JEU"  
1010 PRINT "EST RÉSERVÉ À CEUX"  
1020 PRINT "DONT LE PRENOM"  
1030 PRINT "COMMENCE PAR J"  
1040 STOP
```

Dans ce deuxième exemple, l'ordinateur vous met au défi de le braver.

```

10 PRINT "CE JEU EST EFFRAYANT"
12 PRINT "SEREZ-VOUS ASSEZ BRAVE"
14 PRINT "POUR ATTAQUER LE MONSTRE"
15 PRINT "VERT ET POILU ?"
16 INPUT IS
S ZX17 IF IS (1) = "0" THEN GOTO 20
★ ■ ▲ ● 17 IF LEFT$(IS, 1) = "0" THEN GOTO 20
18 PRINT "LACHE"
19 STOP

```

Pour combiner ces deux exemples, il vous faudra renuméroter la série d'instructions données aux pages précédentes (lignes 11 à 17 à transformer en lignes 20-26). Le programme principal démarrera en 30 et sera suivi du ou des sous-programmes existants.

Voulez-vous jouer encore une fois ?

Pour ne pas avoir à relancer le programme après chaque jeu (RUN), si vous ajoutez les lignes suivantes juste avant l'instruction de fin de programme (STOP ou END), l'ordinateur vous proposera de rejouer avec lui.

```

1000 PRINT "UN AUTRE TOUR ?"
1010 INPUT IS
S ZX1020 IF IS (1) = "0" THEN RUN
★ ■ ▲ ● 1020 IF LEFT$(IS, 1) = "0" THEN RUN
1030 PRINT "Bon - Salut..."
1040 STOP

```

Modifiez la numérotation de cet exemple en fonction de votre programme.



Pour produire des effets sonores

BBC, VIC20, ZX Spectrum et certains Apples peuvent produire des effets sonores.

Placez les instructions correspondantes aux endroits appropriés pour recréer le bruit d'une explosion ou bien jouer un « petit air » en cas de victoire. Dans la mesure où

chaque ordinateur fait appel à des instructions différentes, reportez-vous au manuel de votre appareil. Dans certains cas une seule ligne d'instructions suffira*, dans d'autres cas il faudra créer un sous-programme. Placez GOSUB à l'endroit voulu, numérotez chaque ligne du sous-programme et placez RETURN à la fin. Dans les dernières pages du manuel VIC vous trouverez quelques exemples destinés à produire des effets sonores tels que le « rayon laser », l'« explosion » et l'« alerte rouge ».

* Voici, par exemple, l'effet « coup de feu » sur BBC ; trouvez le meilleur emplacement possible dans votre programme et numérotez la ligne d'instructions :

SOUND 0, - 15, 5, 10.



Note aux utilisateurs de BBC et ZX Spectrum

Utilisateurs de BBC et ZX Spectrum, vous trouverez que certains des jeux présentés dans cet ouvrage ont un déroulement trop rapide. C'est pourquoi des encadrés vous indiquent à chaque fois comment en modifier la vitesse d'exécution. Pour ralentir, utilisez des nombres plus grands. Les derniers modèles BBC, deux fois plus rapides que leurs aînés, rendent ces jeux impossibles à utiliser tels quels. Il faudra intervenir de façon importante sur le nombre conditionnant la vitesse d'exécution des programmes.

Tableau de conversion

Ce tableau montre quelques-unes des variations de langage BASIC pour les ordinateurs cités dans cet ouvrage. Il ne prend pas en compte les instructions graphiques, ni celles produisant des effets sonores ou des couleurs ; les différences sont trop importantes. Constatez également que si la majorité des ordinateurs (sauf le BBC) utilisent les instructions PEEK et POKE, par contre, les systèmes d'adressage en mémoire, c'est-à-dire les nombres placés après celles-ci et permettant de les retrouver, doivent être modifiés pour chacun d'entre eux.

Micro-ordinateur	BBC	VIC/Pet	Apple	TRS-80	ZX Spectrum	ZX81
Tire au sort un nombre compris entre 0 et 0,99999999	RND(1)	RND(1)	RND(1)	RND(0)	RND	RND
Tire au sort un nombre compris entre 1 et N	RND(N)	RND(1)*N + 1	RND(1)*N + 1	RND(N)	RND*N + 1	RND*N + 1
Tire au sort une lettre comprise entre A et Z	CHR\$(RND(26) + 64)	CHR\$(INT(RND(1)*26 + 65))	CHR\$(INT(RND(1)*26 + 65))	CHR\$(RND(26) + 64)	CHR\$(INT(RND*26 + 65))	CHR\$(INT(RND*26 + 30))
Nettoie l'écran	CLS	PRINT CHR\$(147)	HOME	CLS	CLS	CLS
Vérifie si une touche du clavier est pressée	INKEY\$(N)	GET XS	XS = "" IF PEEK(-16384) > 127 THEN GET XS	INKEYS	INKEYS	INKEY\$
Convertit les caractères en nombre code	ASC("X") (utilise le code ASCII)	ASC("X") (utilise le code ASCII)	ASC("X") (utilise le code ASCII)	ASC("X") (utilise le code ASCII)	CODE("X") (utilise le code ASCII)	CODE("X") (utilise le code ZX81)
Déplace le curseur vers le haut	PRINT CHR\$(1)	PRINT CHR\$(145)	CALL - 998	PRINT CHR\$(27)	PRINT CHR\$(11)	PRINT CHR\$(112)
Déplace le curseur vers le bas	PRINT CHR\$(10)	PRINT CHR\$(17)	PRINT CHR\$(10)	PRINT CHR\$(26)	PRINT CHR\$(10)	PRINT CHR\$(113)
Déplace le curseur vers la gauche	PRINT CHR\$(8)	PRINT CHR\$(57)	PRINT CHR\$(8)	PRINT CHR\$(24)	PRINT CHR\$(8)	PRINT CHR\$(114)
Déplace le curseur vers la droite	PRINT CHR\$(9)	PRINT CHR\$(29)	PRINT CHR\$(21)	PRINT CHR\$(25)	PRINT CHR\$(9)	PRINT CHR\$(115)
Prend les N premiers caractères*	LEFT\$(A\$,N)	LEFT\$(A\$,N)	LEFT\$(A\$,N)	LEFT\$(A\$,N)	AS\$(0 TO N)	AS\$(0 TO N)
Prend les N derniers caractères*	RIGHT\$(A\$,N)	RIGHT\$(A\$,N)	RIGHT\$(A\$,N)	RIGHT\$(A\$,N)	ASIN TO)	ASIN TO)
Prend N caractères au milieu*	MID\$(A\$,N1,N2)	MID\$(A\$,N1,N2)	MID\$(A\$,N1,N2)	MID\$(A\$,N1,N2)	ASIN1 TO N2)	ASIN1 TO N2)

* de la chaîne de caractères (variable alphanumérique).



Solutions aux casse-tête

Vos réponses seront peut-être différentes de celles indiquées ci-dessous. Peu importe si cela fonctionne ; regardez cependant si vos solutions sont aussi simples et précises que celles que nous vous proposons.

Page 5

Opération robots

Les instructions de la ligne 90 indiquent le nombre de boucles, donc de possibilités de réponses, que fera l'ordinateur. Plus elles sont nombreuses, meilleure sera votre chance de réussite ; pour rendre le jeu plus facile, il suffira donc d'augmenter la valeur du dernier nombre de la ligne 90. Faites l'inverse pour réduire le nombre de chances.

Page 7

Le Message codé

Dans ce programme, une boucle d'attente est générée lignes 150 et 160. Pendant ce temps, l'ordinateur ne fait rien et ne passe à l'instruction suivante (annulation de l'affichage du message) qu'à la fin de cette boucle.

Si vous souhaitez que le message vital reste affiché plus longtemps, augmenter le nombre de tour de cette boucle en modifiant, en l'augmentant, la valeur du dernier nombre de la ligne 150.

Page 9

Duel

Pour que l'ordinateur puisse également rater son tir, ajoutez les lignes suivantes à celles figurant dans l'encadré « pour rendre le jeu plus difficile » page 8 :

```
155 IF RND < .1 THEN GOTO 250
250 PRINT "IL A TIRE, MAIS A COTE"
260 GOTO 90
```

Selon votre appareil, il faudra lire, ligne 155, RND (1) ou RND (0).

Page 11

Les Blindés attaquent !

En ajoutant au programme les quelques lignes ci-dessous, les robots pourront vous apercevoir.

```
175 IF RND < .05 THEN GOTO 250
250 PRINT "ILS VOUS ONT VU -
AAGGGHHHHH..."
260 STOP
```

Ligne 175, l'instruction RND devra être adaptée à votre appareil.

Vous pouvez donner au nombre 05 toutes valeurs comprises entre 0 et 0,999, mais rappelez-vous simplement que plus la valeur est grande, plus les robots ont une chance de vous apercevoir.

Page 13

Le siège du château fort

Il s'agissait d'un casse-tête difficile et rien de plus normal si vous n'avez pas trouvé la solution. Essayez tout de même notre proposition pour obtenir deux cibles.

```
55 LET P = INT (RND + .5) * 4 + 1
S ZX57 LET PS = CHR$ (P + 51)
* ■ ▲ ● 57 LET PS = CHR$ (P + 78)
70 IF L = T THEN LET RS = RS + PS
180 LET S = S + P
```

Page 15

Les envahisseurs

Pour modifier le score et donner 100 points à U, V, W, X et Y et 10 points aux autres, effectuez les modifications suivantes :

```
220 LET H = H + 10
225 IF PS > "T" AND PS < "Z" THEN LET H = H + 90
```

Page 17

L'arme secrète

Pour compléter le score, tel que proposé dans le casse-tête, placez les instructions suivantes :


JEUX ELECTRONIQUES

Voulez-vous jouer avec un micro-ordinateur ? Ce livre vous offre de nombreux programmes de jeux passionnants, accompagnés d'explications claires et détaillées, de suggestions pour les modifier, de casse-tête à résoudre. En jouant à ces jeux, même un débutant comprendra très vite comment fonctionne un programme, et il n'aura bientôt plus qu'un désir : réaliser ses propres programmes. Pour l'y aider, il trouvera dans ce livre des conseils et des « trucs », un lexique de Basic, ainsi qu'une table de conversion qui permet d'adapter à tous les ordinateurs les programmes proposés dans les revues ou magazines.

Les programmes de ce livre sont réalisables sur les ordinateurs suivants : ZX81, BBC, TRS-80, Pet, les Apples qui utilisent le Palsoft Basic, et le ZX Spectrum.



Dans la même collection :



25,00 FF TTC



9 782010 093654

ECHOS
ELECTRONIQUE

29/0333/
83-V
Imprimé en Espagne
par Edit. Elepuru, S.A.L. - Zamudio (Vizcaya)